Exercice du 2 avril

```
> with(LinearAlgebra):
> A := <<0,3,4> | <-20,27,11> | <-14,-4,-2>>;
```

$$A := \begin{bmatrix} 0 & -20 & -14 \\ 3 & 27 & -4 \\ 4 & 11 & -2 \end{bmatrix} \quad (1)$$

Q et R calculées par la fonction QRDecomposition

```
> QRDecomposition (A);
```

$$\begin{bmatrix} 0 & -\frac{4}{5} & -\frac{3}{5} \\ \frac{3}{5} & \frac{12}{25} & -\frac{16}{25} \\ \frac{4}{5} & -\frac{9}{25} & \frac{12}{25} \end{bmatrix}, \begin{bmatrix} 5 & 25 & -4 \\ 0 & 25 & 10 \\ 0 & 0 & 10 \end{bmatrix} \quad (2)$$

Première itération

```
> x := A[1..3,1];
```

$$x := \begin{bmatrix} 0 \\ 3 \\ 4 \end{bmatrix} \quad (3)$$

```
> v := x + <5,0,0>;
```

$$v := \begin{bmatrix} 5 \\ 3 \\ 4 \end{bmatrix} \quad (4)$$

```
> Q1 := IdentityMatrix(3) - 2*v.Transpose(v)/(Transpose(v).v);
```

$$Q1 := \begin{bmatrix} 0 & -\frac{3}{5} & -\frac{4}{5} \\ -\frac{3}{5} & \frac{16}{25} & -\frac{12}{25} \\ -\frac{4}{5} & -\frac{12}{25} & \frac{9}{25} \end{bmatrix} \quad (5)$$

```
> Q1A := Q1 . A;
```

(6)

$$Q1A := \begin{bmatrix} -5 & -25 & 4 \\ 0 & 24 & \frac{34}{5} \\ 0 & 7 & \frac{62}{5} \end{bmatrix} \quad (6)$$

```
> v := <49,7>;
```

$$v := \begin{bmatrix} 49 \\ 7 \end{bmatrix} \quad (7)$$

```
> F := IdentityMatrix(2) - 2*v.Transpose(v) / (Transpose(v).v);
```

$$F := \begin{bmatrix} -\frac{24}{25} & -\frac{7}{25} \\ -\frac{7}{25} & \frac{24}{25} \end{bmatrix} \quad (8)$$

```
> F . Q1A[2..3,2..3];
```

$$\begin{bmatrix} -25 & -10 \\ 0 & 10 \end{bmatrix} \quad (9)$$

On n'a pas trouvé exactement le même résultat parce qu'on ne s'est pas assuré que les signes des éléments diagonaux de R étaient positifs. Pour préparer une résolution en Python, choisissons un vecteur x

```
> x := < 4, 2, -5>;
```

$$x := \begin{bmatrix} 4 \\ 2 \\ -5 \end{bmatrix} \quad (10)$$

On en déduit un vecteur b.

```
> b := A . x;
```

$$b := \begin{bmatrix} 30 \\ 86 \\ 48 \end{bmatrix} \quad (11)$$

Calculons la matrice Q qui correspond aux calculs que nous avons faits en TD.

```
> Q2 := <<1,0,0> | <<0 | 0>, F>>;
```

$$Q2 := \begin{bmatrix} 1 & 0 & 0 \\ 0 & -\frac{24}{25} & -\frac{7}{25} \\ 0 & -\frac{7}{25} & \frac{24}{25} \end{bmatrix} \quad (12)$$

```
> Q := Transpose(Q2 . Q1);
```

$$Q := \begin{bmatrix} 0 & \frac{4}{5} & -\frac{3}{5} \\ -\frac{3}{5} & -\frac{12}{25} & -\frac{16}{25} \\ -\frac{4}{5} & \frac{9}{25} & \frac{12}{25} \end{bmatrix} \quad \mathbf{(13)}$$

Et la matrice R

```
> R := Q2 . Q1 . A;
```

$$R := \begin{bmatrix} -5 & -25 & 4 \\ 0 & -25 & -10 \\ 0 & 0 & 10 \end{bmatrix} \quad \mathbf{(14)}$$

On vérifie que, pour résoudre A . x = b, il suffit de résoudre R . x = Q^T . b

```
> Transpose (Q) . b;
```

$$\begin{bmatrix} -90 \\ 0 \\ -50 \end{bmatrix} \quad \mathbf{(15)}$$

```
> LinearSolve (R, Transpose (Q) . b);
```

$$\begin{bmatrix} 4 \\ 2 \\ -5 \end{bmatrix} \quad \mathbf{(16)}$$